

***Simulation of
Switching Power Supply
Performance Using the
Personal Computer***

by Lloyd H. Dixon, Jr.

Topic 3

SIMULATION OF SWITCHING POWER SUPPLY PERFORMANCE USING THE PERSONAL COMPUTER

Lloyd H. Dixon, Jr

The ubiquitous personal computer is capable of simulating switching power supply performance using simple programs written in BASIC interpreter language. Performing integration by parts, the computer calculates small changes in current and voltage in each circuit element, working its way through the entire circuit and iterating around this loop many times in a single switching cycle. The only mathematics required is simple algebra. Complex equivalent circuit models are completely avoided. The actual circuit is exercised in the computer just as it would be on the lab bench. Successful simulation depends only on how faithfully the computer version of the circuit conforms to the actual power supply circuit.

Benefits and Limitations:

This technique provides a method of "software breadboarding" for checking out the design of any switching power supply. Not an artificially derived model, the computer program simulates operation of the actual circuit. Small signal loop stability and large signal behavior are observed. The circuit can be exercised almost effortlessly by the computer under a wide range of conditions, including startup and overload. Using an interpreter language permits the program to be interrupted and any values may be changed before continuing program execution. This is useful in simulating step changes in load current or line voltage.

This method is not intended to supplant a strong initial design effort, but to provide a fast and accurate method to check out and optimize the design. It provides excellent visibility into unforeseen problems and facilitates a more knowledgeable approach to solving these problems much more easily than in the lab. As a debugging tool, it is a simple matter to change the displayed variables or add new ones, then rerun the program to more clearly define the problem and its solution. Behavior under operating condition extremes is easily checked out.

Neither is this method a substitute for final lab checkout, although it does minimize the need for hardware debugging. When the computer simulation checks out with the design expectation and the final lab tests confirm it, there is a high probability of success.

The very act of writing the few BASIC program lines for the circuit being evaluated provides much insight into its operation. The implied question "Do I have the circuit correctly captured in the program?" often uncovers problems that were initially overlooked.

On the IBM PC, the running time for 1 cycle at the switching frequency is typically 4 seconds. Depending on the topology and L-C filter values of the specific application, it may take 1/2 to 5 minutes to run from startup until equilibrium operation is reached. The running time is just about right to permit the designer to comfortably observe its progress, noting whether it

conforms to expectations and attempting to understand why any deviations occur. Execution continues indefinitely until it is stopped by keyboard interruption (CTRL-BREAK on the PC).

The simulation program as presented makes 20 to 50 iterations per switching period and provides an accurate representation of circuit behavior at or below the switching frequency. It is obviously incapable of dealing with higher frequency phenomena such as noise spikes and leakage inductance effects, although these could conceivably be handled by smaller iteration intervals and increased circuit/program complexity which would greatly slow down execution time.

The continuous mode buck regulator with simple duty cycle control used as the program example herein can serve as the basis for any other power supply circuit by changing a few program lines.

Program Features:

A brief overview of the program listing printed at the end of this paper:

Lines 1000-1280 set up miscellaneous definitions including title strings and a menu providing the choices of viewing the program conditions, running the program, and directing output to the printer. Except for the title in line 1000, these lines need not be changed for different applications.

Line 2010 defines the data column headings that are displayed at the top of each page. Lines 2020-2060 display the initial RUNtime title and data column headings. Lines 2080-2250 are subroutines which output the desired data, in this case twice during each switching period -- when the power switch turns on and when it turns off.


Lines 2260-2400 define circuit parameter variable names and initial values in a set of READ and DATA statements. Any of these values may be changed before running the program by editing the DATA lines. If the program is then SAVEd, the changed values will become the initial values the next time the program is loaded. When the menu choice is made to display operating conditions, lines 1240 and 1250 accomplish this by LISTing lines 2260-2400. Thus it is not necessary to duplicate variable names for displaying the operating conditions, making it easier to change these variable names for different applications. When it is desired to change a parameter value, this technique facilitates the process by showing the actual lines to be edited along with their line numbers.

Lines 2410-2500 define other initial values. Load resistance, ROUT, is calculated from the specified output voltage/current, VO/IO. Counters and other conditions are also set up. Note line 2490 defining resistor RD which divides the 5 volt output down to the 2 volt reference level, VNI (see Fig. 2). RD has no effect on the loop gain. RD is defined in this way so that it will be adjusted automatically to maintain the desired output voltage if RIP or RIZ are changed before running the program. RD could be defined in the parameter listings, instead.

The heart of the program is the computation section, lines 3000-3440. This section is most uniquely related to the specific circuit, yet perhaps half of the lines could be used without change in most applications.

Throughout this program, frequency is expressed in MHz, time in μs , inductance in μH , and capacitance in μF , in order to avoid the repeated use of 10^{-6} , thus simplifying the equations.

The Power Switching Circuit -- Figure 1:



The power switching circuit and filter are shown in Figure 1. Component values are listed at the beginning of the sample run at the end of this section. For the values given, the L-C filter resonance is at 2770 Hz, contributing a 2-pole second order characteristic above that frequency. One of these poles is compensated by the tantalum capacitor's ESR zero which occurs above 21,200 Hz. The gain from the pulse width modulator control terminal to the output of the supply at low frequencies is 8.15, or 18.2 dB. The control to output gain is .118, or -18.5 dB at 25 kHz, which is the intended crossover frequency (1/4 the switching frequency).

The program branches to line 3040 on the first iteration of each switching cycle. Cycle counter CCNT is incremented, iteration interval TI is initially set to 1/100 of the switching period, and cumulative interval TC is zeroed. The SWITCH is set "on" and VIN1, the voltage at the input of the filter, is set equal to VIN. A forward converter or push-pull version of the buck regulator with transformer coupling would have the identical circuit and equations except VIN1 above would be set equal to VIN/N - VD (Primary side Vin divided by turns ratio minus rectifier drop). On successive iterations, the program will branch to line 3130 until it is time for the switch to turn off.

When the comparator output changes state to turn the switch off, the program branches to line 3090, VIN1 is set to -VD (the circuit free-wheels through the rectifier), and the iteration interval is made exactly 1/10 of the time remaining to the end of the switching period. Thus there will be exactly 10 iterations during the "off" time which helps reduce the running time. Except for initial turn on and turn off, the program jumps to line 3130 at the beginning of each iteration.

The cumulative time interval TC within the current period is updated by line 3130. Line 3140 updates the filter inductor current. The new IL value equals the old IL plus ΔIL , which is $V_L \cdot \Delta T / L$. The voltage across the inductor is (VIN1 - VOUT), and ΔT is the iteration interval, TI. VIN1 equals VIN during the "on" time, and equals -VD, the freewheeling diode drop, during the "off" time. To greatly simplify the calculation, an important assumption is made: that VOUT, at the inductor output, is constant during each iteration. This assumption may not be valid in an abstract theoretical sense, but in a practical power supply it is certainly acceptable. The VOUT


variation over the entire switching period (the output ripple voltage) is typically only 1-2%, so it will change less than 0.1% during the iteration interval. Any small error that occurs will not cumulatively increase, but self-corrects during the next iteration.

The second statement in line 3140 is used when the inductor current becomes discontinuous, i.e. the current is zero during a portion of each switching period. This occurs whenever load current drops below the critical inductor current level. In the actual circuit, the rectifier prevents the current from going negative. In the program, the second part of 3140 performs the same function.

Line 3150 computes IC from the new IL and old VOUT values, 3160 computes the new VC, and 3170 calculates the new VOUT value based on the new VC and new IL values (changes in VC and VOUT are very small for each iteration).

Line 3180 finds the minimum and maximum values of VOUT which define the peak-to-peak ripple. This may occur at any time during the switching cycle, not necessarily at switch on and switch off. The determination is made by looking for a change in sign of the ΔV_{OUT} between the present and the previous iteration. The subroutine at line 2180 prints these min/max VOUT values to the right of the normal data columns, including the time within the cycle that they occur.

The Error Amplifier/Compensation Network -- Figure 2:



This remaining portion of the power supply using a UC3524 control IC is shown in Figure 2. A low frequency pole provides high DC gain. Two zeros come in at 2770 Hz to compensate the two filter resonance poles. A single pole compensates the ESR zero keeping the overall loop gain trending ever downward with a net single pole characteristic to well above the crossover frequency. The gain of the error amplifier with its compensation network is +18.5 dB at 25 kHz and above, so that the overall loop gain including the power circuit is 0 dB at 25 kHz, the crossover frequency. The error amplifier gain is also +18.5 dB at the 100 kHz switching frequency. This amplifies the 0.1 volt ripple component of VOUT to 0.85 volts at the output of the error amplifier. It is then applied to the control terminal of the modulator and compared against the sawtooth ramp whose amplitude is 2.7 volts. If the amplified ripple exceeds the sawtooth amplitude (which would happen with a higher crossover frequency), the circuit becomes unstable. In this case, the loop gain has been pushed about as high as it should, and the circuit response is excellent as shown by the demonstration run.

The computer iterates its way through the error amplifier circuit in lines 3220-3280. A simplifying assumption is made that the circuit gain-bandwidth envelope is determined completely by the external components, i.e. the small signal gain-bandwidth of the error amplifier is so high it is negligible. This is indeed what should be expected, otherwise the poles and zeros of the compensation circuit cannot function as intended. Always compare the Bode plot of the overall compensation circuit against the error amplifier alone to make certain it is up to the task. So the error amplifier is assumed to be ideal except for its large signal limitations which will be discussed shortly.

Lines 3220 and 3230 presume the error amplifier is functioning normally, in which case the voltage at the inverting input is essentially equal (within a fraction of a millivolt) to the 2 V reference voltage on the non-inverting input, VNI. Using the old values of VCI and VCF, the input and feedback currents II and IFB are calculated, and error amplifier output voltage VCTL (which is applied to the control terminal of the modulator) is calculated in line 3240.

Before proceeding to update the capacitor voltages VCI and VCF, lines 3250 and 3260 check whether VCTL has violated the E/A output clamp voltages or whether IFB has exceeded the E/A output current source or sink capability. If so, the program jumps to the subroutines at 3350 and/or 3400. The assumption of normal operation is abandoned, and the output is constrained by either the voltage clamp or current limits. The error amplifier gain is temporarily zero in this condition and the non-inverting input voltage is not maintained equal to VNI. II and IFB are recalculated based on the appropriate situation in lines 3360-3370 or 3410-3420. This is done so that even during "abnormal" circuit operation the voltages on VCI and VNI will be correctly updated. These equations for recalculating II and IFB and VCTL were developed by algebraically combining several simpler sequential steps. There is no need to retain any intermediate values and this speeds up program execution. VCI and VCF are updated in 3270 and 3280 based on the final current values.

If the power switch is off, 3290 looks for the end of the switching period and when it occurs, causes the data to be viewed/printed and then branches to 3040, the beginning of the next cycle.

If the switch is on, 3300 checks the comparator input and the current limit to determine whether to turn the switch off. If so, data is viewed/printed and the program branches to 3090, the beginning of the "off" period. If the switch is on and the end of the "on" period has not yet been reached, 3310 makes an estimate on how close it is to turn-off and adjusts the iteration interval accordingly. The interval needs to be small as the end of the "on" time approaches in order to terminate the "on" time with adequate precision. If the program always used this small increment it would run very slowly. So line 3310 speeds up execution time considerably.

Techniques to Improve Speed and Accuracy:

In execution time of the BASIC interpreter program in this application depends, in order of importance, on:

1. Each variable must be looked up each time it is encountered (even in the same equation) in the variables table maintained by BASIC. The look-up time depends on how many variables are in the entire program, which determines the length of the table. It is worth the effort to minimize the number of variables encountered in the iteration loops.
2. Of next importance is the time BASIC spends interpreting each statement. This can be minimized by combining statements algebraically, omitting intermediate variables. It is only necessary to update and save (as a named variable) capacitor voltages, inductor currents, and those values which may be viewed/printed as data. All other values can be eliminated if possible by combining equations. This also minimizes variable look-up time as above.
3. Computation time is the least important in this application because the calculations are so simple: +, -, /, and *.

With one exception, the iteration interval can be quite large -- 1/10 to 1/20 of the switching period -- without impairing accuracy. This means 10 or 20 iterations per cycle which will run at a good speed. Try running the same program with the iteration interval changed. If the results are close to each other use the larger increment which will run faster. However, using a long interval throughout results in insufficient accuracy in defining the length of the "on" time, which must always terminate at an interval boundary. If the duty ratio is supposed to be .31, but there are only 10 iterations per cycle, the switch will turn off at .4, which is a great deal of overshoot. An interval of 1/100 is more appropriate when approaching the turn-off point, but is too slow for general use. The only solution is to use some method of estimating proximity to turn-off and using a fine increment when it gets close. This is probably the biggest source of error in using this simulation method.

Unlike the buck regulator, flyback equivalent circuits are very different in the "on" state compared to the "off" state. In writing a single program section to cover both states, there will be several lines that apply to one state and not the other, and several IF....THEN statements to direct program execution to the proper lines. All this slows up program execution, and makes writing the program more difficult. In such cases, it is better to use two completely separate program segments, one for each state. Switching to the proper segment is done no more than once per iteration. This will slightly increase the total program length, but it will run faster.

```

2350 ' Modulator:
2360 READ VSMIN, VSMAX, VCLAMP
2370 DATA 0.8, 3.5, 2.2
2380 '
2390 ' To change values, edit DATA statements above. SAVE program (optional)
2400 ' Then RUN again.
2410 '
2420 ' Initial values
2430 T=1/F:' Period
2440 TC=0:' Cumulative interval (time within cycle)
2450 CCNT=0:' Cycle count
2460 ROUT=VO/IO:' Load resistance
2470 IF VNI=VO THEN RD=1E+10 ELSE RD=VNI*(RIP+RIZ)/(VO-VNI):' Divider Resistor
2480 VS=VSMAX-VSMIN:' Ramp amplitude
2490 VD=.6:' Rectifier drop
2500 '
3000 ' COMPUTATION
3010 '
3020 ' Power Switch ON
3030 '
3040 CCNT=CCNT+1:TI=T/100:TC=0:SWITCH=1:' Switch ON, Begin new cycle
3050 VIN1=VIN:GOTO 3130
3060 '
3070 ' Power switch OFF
3080 '
3090 TI=(T-TC)/10:SWITCH=0:' Switch OFF
3100 VIN1=-VD
3110 '
3120 ' Power switch and filter
3130 TC=TC+TI
3140 IL=IL+(VIN1-VOUT)*TI/L:IF IL<0 THEN IL=0
3150 IC=IL-VOUT/ROUT
3160 VC=VC+IC*TI/C
3170 VOUT=VC+IC*ESR
3180 M3=M2:M2=M1:M1=VOUT:IF SGN(M2-M3)<>SGN(M1-M2) THEN GOSUB 2180
3190 '
3200 'E/A Type 2
3210 '
3220 II=(VOUT-VNI-VCI)/RIP
3230 IFB= II - VNI/RD
3240 VCTL=VNI-VCF-IFB*RFZ
3250 IF VCTL>VCLAMP OR VCTL<0 THEN GOSUB 3350
3260 IF IFB>ISNK OR IFB<-ISRC THEN GOSUB 3400
3270 VCI=VCI+(II-VCI/RIZ)*TI/CI
3280 VCF=VCF+IFB*TI/CF
3290 IF SWITCH=0 THEN IF TC>.999*T THEN GOSUB 2090:GOTO 3040 ELSE GOTO 3130
3300 IF VCTL<=(VS*TC/T+VSMIN) OR IL>ILIM THEN GOSUB 2090:GOTO 3090
3310 IF VCTL-(VS*TC/T+VSMIN)<.2 OR IL>.9*ILIM THEN TI=T/100 ELSE TI=T/20
3320 GOTO 3130
3330 '
3340 ' Vctl clamped
3350 IF VCTL>VCLAMP THEN VCTL=VCLAMP ELSE VCTL=0
3360 IFB=((VOUT-VCI)*(1-RIP/(RD+RIP))-VCF-VCTL)/(RFZ+RD*RIP/(RD+RIP))
3370 II=(VOUT-VCI+IFB*RD)/(RD+RIP)
3380 RETURN
3390 ' Ifb too great
3400 IF IFB<-ISRC THEN IFB=-ISRC ELSE IFB=ISNK
3410 II=(VOUT-VCI+IFB*RD)/(RD+RIP)
3420 VCTL=VOUT-II*RIP-VCI-IFB*RFZ-VCF
3430 RETURN
3440 END

```



```

1000 T2$="BUCKDC1 -- Buck, Continuous, Duty Cycle Control"
1010 T1$="SWITCHING POWER SUPPLY PERFORMANCE SIMULATION"
1020 ' L. H. Dixon UNITRODE CORP. 1/15/85
1030 '
1040 'System definitions:
1050 PLEN=66;TLEN=24: Printer, Terminal #lines/pg
1060 LF$=CHR$(13)+CHR$(10): 1 line feed
1070 LF2$=LF$+LF$: 2 " "
1080 FF$=CHR$(12): Form Feed
1090 TITL$=T1$+DATE$+", "+TIME$+LF2$+T2$+LF$
1100 '
1110 PRINT FF$+TITL$+LF$
1120 PRINT " 1 Display/change operating conditions
1130 PRINT " 2 Run simulation on terminal
1140 PRINT " 3 Print operating conditions
1150 PRINT " (First make sure printer is at page top)
1160 PRINT " 4 Run simulation with printout
1170 PRINT
1180 PRINT" CTRL/BREAK to terminate.
1190 PRINT LF2$
1200 PRINT " Enter 1, 2, 3, or 4: _"+CHR$(29);
1210 '
1220 S$=INKEY$:IF S$="" THEN 1220 ELSE PRINT S$
1230 '
1240 IF S$="1" THEN PRINT FF$+TITL$:LIST 2260-2400
1250 IF S$="3" THEN PRINT FF$+LF2$+"RUN again":LPRINT TITL$:LLIST 2260-2380
1260 ' STOP is inherent after execution of LIST statements in the above lines
1270 IF S$<>"2" AND S$<>"4" THEN 1200
1280 '
2000 ' Print column headings
2010 HDR$="Cycle Time IL IC Vout "
2020 HDR$=HDR$+DATE$+", "
2030 IF S$="4" THEN LPRINT LF$+TITL$ +HDR$+TIME$ ELSE PRINT FF$+TITL$
2040 PRINT LF$+HDR$+TIME$
2050 TCNT=0:PCNT=20:' Init terminal, printer line count
2060 GOTO 2260
2070 '
2080 ' Display/print results
2090 PRINT:TCNT=TCNT+1:IF TCNT)=TLEN-2 THEN PRINT HDR$+TIME$:TCNT=0
2100 PRINT USING "####";CCNT;
2110 PRINT USING "#####.##";TC/T;IL;IC;VOUT;
2120 IF S$<>"4" THEN RETURN
2130 LPRINT:PCNT=PCNT+1:IF PCNT>PLEN-10 THEN LPRINT FF$+HDR$+TIME$+LF$:PCNT=0
2140 LPRINT USING "####";CCNT;
2150 LPRINT USING "#####.##";TC/T;IL;IC;VOUT;
2160 RETURN
2170 '
2180 ' Max/min values
2185 IF CCNT=1 THEN RETURN
2190 PRINT USING " #####.##";(TC-TI)/T;
2200 PRINT USING "####.##";M2;
2210 IF S$<>"4" THEN RETURN
2220 LPRINT USING " #####.##";(TC-TI)/T;
2230 LPRINT USING "####.##";M2;
2240 RETURN
2250 '
2260 '
2270 ' Power Circuit Parameters:
2280 READ F, VIN, VO, IO, ILIN, L, C, ESR
2290 DATA 0.1, 16, 5, 20, 25, 11, 300, .025
2300 '
2310 ' Error Amplifier Values:
2320 READ VNI, RIP, RIZ, CI, RFZ, CF, ISRC, ISNK
2330 DATA 2.0, 4504, 30000, .0019, 35000, .0015, .0001, .0002
2340 '

```

SWITCHING POWER SUPPLY PERFORMANCE SIMULATION

02-07-1985, Cycle Time

BUCKDC1 -- Buck, Continuous, Duty Cycle Control	19	0.31	22.57	2.51	5.02
	19	1.00	19.06	-0.79	4.95
2260 ' Power Circuit Parameters:	20	0.31	22.16	2.09	5.03
2270 ' F, VIN, VO, IO, ILIM, L, C, ESR	20	1.00	18.65	-1.21	4.95
2280 READ	21	0.33	21.95	1.82	5.04
2290 DATA 0.1, 16, 5, 20, 25, 11, 300, .025	21	1.00	18.54	-1.34	4.96
2300 ' Error Amplifier Values:	22	0.34	21.94	1.78	5.04
2310 ' VNI, RIP, RIZ, CI, RFZ, CF, ISRC, ISNK	22	1.00	18.57	-1.33	4.97
2320 READ	23	0.33	21.87	1.70	5.05
2330 DATA 2.0, 4504, 30000, .0019, 36000, .0015, .0001, .0002	23	1.00	18.45	-1.46	4.97
2340 ' Modulator:	24	0.34	21.85	1.67	5.05
2350 ' VSMIN, VSMAX, VCLAMP	24	1.00	18.48	-1.44	4.97
2360 READ	25	0.33	21.78	1.60	5.05
2370 DATA 0.8, 3.5, 2.2	25	1.00	18.36	-1.55	4.97
2380 '	26	0.34	21.76	1.56	5.05
	26	1.00	18.39	-1.52	4.97

SWITCHING POWER SUPPLY PERFORMANCE SIMULATION

02-07-1985,

INT- ROJT .25 → 1.04

BUCKDC1 -- Buck, Continuous, Duty Cycle Control
Cycle Time IL IC Vout 02-07-1985

1	0.52	7.52	6.62	0.23	27	0.31	21.49	16.46	5.44
1	1.00	7.14	5.92	0.31	27	1.00	17.64	12.01	5.65
2	0.52	14.49	12.01	0.63	28	0.31	17.73	12.08	5.66
2	1.00	13.92	10.86	0.78	28	1.00	12.03	6.23	5.80
3	0.52	21.04	16.42	1.16	29	0.01	12.12	6.32	5.81
3	1.00	20.23	14.82	1.37	29	1.00	6.36	0.58	5.77
4	0.37	25.10	18.46	1.67	30	0.01	6.45	0.68	5.77
4	1.00	23.72	15.97	1.97	30	1.00	0.79	-4.79	5.56
5	0.11	25.12	16.95	2.05	31	0.08	1.55	-4.01	5.56
5	1.00	22.84	13.34	2.41	31	1.00	0.00	-5.38	5.37
6	0.18	25.06	14.96	2.53	32	0.18	1.74	-3.64	5.38
6	1.00	22.62	11.50	2.80	32	1.00	0.00	-5.22	5.21
7	0.20	25.01	13.31	2.93	33	0.24	2.35	-2.88	5.23
7	1.00	22.37	9.84	3.15	33	1.00	0.00	-5.08	5.06
8	0.23	25.04	11.92	3.29	34	0.29	2.88	-2.22	5.10
8	1.00	22.26	8.49	3.46	34	1.00	0.00	-4.95	4.94
9	0.25	25.10	10.74	3.59	35	0.35	3.51	-1.48	4.99
9	1.00	22.19	7.31	3.73	35	1.00	0.25	-4.61	4.84
10	0.26	25.08	9.65	3.86	36	0.38	4.09	-0.81	4.91
10	1.00	22.04	6.20	3.97	36	1.00	1.02	-3.78	4.78
11	0.28	25.09	8.70	4.10	37	0.39	4.98	0.13	4.86
11	1.00	21.98	5.28	4.18	37	1.00	1.98	-2.79	4.76
12	0.29	25.08	7.87	4.31	38	0.37	5.75	0.91	4.84
12	1.00	21.89	4.45	4.36	38	1.00	2.66	-2.10	4.75
13	0.30	25.05	7.11	4.49	39	0.36	6.33	1.49	4.84
13	1.00	21.80	3.70	4.53	39	1.00	3.18	-1.59	4.76
14	0.31	25.02	6.44	4.65	40	0.34	6.65	1.80	4.84
14	1.00	21.72	3.04	4.67	40	1.00	3.40	-1.38	4.77
15	0.30	24.79	5.69	4.78	41	0.34	6.86	2.00	4.96
15	1.00	21.37	2.23	4.78	41	1.00	3.60	-1.19	4.78
16	0.28	24.21	4.72	4.88	42	0.33	6.96	2.09	4.87
16	1.00	20.63	1.19	4.85	42	1.00	3.64	-1.17	4.80
17	0.29	23.56	3.79	4.95	43	0.33	6.99	2.11	4.89
17	1.00	19.99	0.34	4.90	43	1.00	3.67	-1.16	4.81
18	0.30	23.00	3.05	4.99	44	0.33	7.01	2.11	4.90
18	1.00	19.46	-0.31	4.93	44	1.00	3.68	-1.16	4.83
					45	0.33	7.02	2.11	4.92
					45	1.00	3.68	-1.17	4.84
					46	0.33	7.01	2.09	4.93
					46	1.00	3.66	-1.20	4.85
					47	0.33	7.00	2.06	4.94

Cycle	Time	IL	IC	Vout	Cycle	Time	IL	IC	Vout	Cycle	Time	IL	IC	Vout
47	1.00	3.64	-1.24	4.87	76	0.52	7.95	-8.38	4.08	104	1.00	18.82	-1.60	5.09
48	0.33	6.97	2.02	4.95	76	1.00	5.95	-9.73	3.90	105	0.25	22.43	1.73	5.18
48	1.00	3.60	-1.28	4.88	77	0.52	9.31	-6.12	3.86	105	1.00	18.51	-1.87	5.08
49	0.33	6.93	1.97	4.96	77	1.00	7.40	-7.51	3.71	106	0.26	22.27	1.65	5.17
49	1.00	3.56	-1.33	4.88	78	0.52	10.84	-3.97	3.70	106	1.00	18.41	-1.93	5.07
50	0.33	6.88	1.92	4.97	78	1.00	8.99	-5.42	3.59	107	0.26	22.17	1.59	5.16
50	1.00	3.51	-1.39	4.89	79	0.52	12.49	-1.98	3.62	107	1.00	18.32	-1.99	5.06
51	0.33	6.83	1.86	4.98	79	1.00	10.66	-3.51	3.53	108	0.26	22.07	1.54	5.15
51	1.00	3.45	-1.45	4.89	80	0.52	14.18	-0.16	3.59	108	1.00	18.23	-2.02	5.05
52	0.33	6.78	1.80	4.98	80	1.00	12.37	-1.78	3.53	109	0.26	21.99	1.51	5.14
52	1.00	3.40	-1.51	4.90	81	0.52	15.88	1.44	3.61	109	1.00	18.16	-2.05	5.04
53	0.34	6.82	1.84	4.98	81	1.00	14.05	-0.27	3.58	110	0.26	21.92	1.49	5.13
53	1.00	3.49	-1.43	4.90	82	0.52	17.54	2.83	3.68	110	1.00	18.10	-2.06	5.03
54	0.33	6.80	1.82	4.99	82	1.00	15.67	1.01	3.66	111	0.26	21.87	1.48	5.11
54	1.00	3.42	-1.50	4.91	83	0.52	19.12	3.99	3.78	111	1.00	18.05	-2.06	5.01
55	0.34	6.84	1.84	4.99	83	1.00	17.20	2.07	3.78	112	0.27	21.96	1.55	5.10
55	1.00	3.50	-1.43	4.91	84	0.52	20.59	4.93	3.92	112	1.00	18.20	-1.89	5.01
56	0.33	6.81	1.82	5.00	84	1.00	18.61	2.90	3.93	113	0.26	21.97	1.65	5.10
56	1.00	3.42	-1.51	4.92	85	0.52	21.93	5.64	4.07	113	1.00	18.16	-1.90	5.00
INT. VIN 16 → 11 V					85	1.00	19.88	3.52	4.09	114	0.26	21.94	1.64	5.09
57	0.34	6.84	1.84	5.00	86	0.52	23.12	6.15	4.24	114	1.00	18.13	-1.90	5.00
57	1.00	3.49	-1.44	4.92	86	1.00	21.00	3.93	4.27	115	0.26	21.91	1.64	5.09
58	0.41	5.75	0.78	4.98	87	0.52	24.14	6.47	4.42	115	1.00	18.11	-1.89	4.99
58	1.00	2.78	-2.12	4.89	87	1.00	21.95	4.16	4.45	116	0.26	21.89	1.65	5.08
59	0.47	5.38	0.44	4.94	88	0.52	25.01	6.61	4.60	116	1.00	18.09	-1.89	4.98
59	1.00	2.73	-2.14	4.86	88	1.00	22.73	4.22	4.63	117	0.26	21.87	1.66	5.07
60	0.50	5.51	0.60	4.92	89	0.40	25.03	6.07	4.74	117	1.00	18.08	-1.87	4.98
60	1.00	2.02	-1.84	4.84	89	1.00	22.11	3.07	4.76	118	0.27	22.00	1.74	5.07
61	0.51	5.86	0.95	4.91	90	0.39	24.31	4.89	4.86	118	1.00	18.26	-1.70	4.98
61	1.00	3.42	-1.43	4.84	90	1.00	21.28	1.88	4.85	119	0.26	22.04	1.84	5.07
62	0.49	6.15	1.24	4.91	91	0.39	23.45	3.74	4.93	119	1.00	18.25	-1.71	4.98
62	1.00	3.61	-1.25	4.84	91	1.00	20.39	0.77	4.90	120	0.26	22.03	1.83	5.07
63	0.48	6.28	1.37	4.91	92	0.41	22.65	2.76	4.97	120	1.00	18.24	-1.71	4.98
63	1.00	3.69	-1.17	4.85	92	1.00	19.67	-0.07	4.93	121	0.26	22.02	1.82	5.07
64	0.48	6.36	1.44	4.92	93	0.43	22.03	2.05	5.00	121	1.00	18.23	-1.72	4.98
64	1.00	3.76	-1.10	4.86	93	1.00	19.14	-0.65	4.94	122	0.26	22.01	1.82	5.07
65	0.47	6.37	1.45	4.92	94	0.45	21.61	1.60	5.00					
65	1.00	3.72	-1.14	4.86	94	1.00	18.82	-0.98	4.94					
66	0.48	6.39	1.46	4.93	95	0.47	21.40	1.37	5.01					
66	1.00	3.79	-1.09	4.87	95	1.00	18.71	-1.11	4.95					
67	0.47	6.40	1.46	4.94	96	0.48	21.34	1.30	5.01					
67	1.00	3.74	-1.14	4.87	96	1.00	18.70	-1.13	4.95					
68	0.48	6.40	1.46	4.94	97	0.48	21.33	1.28	5.01					
68	1.00	3.80	-1.09	4.88	97	1.00	18.68	-1.15	4.95					
69	0.48	6.45	1.51	4.95	98	0.48	21.31	1.26	5.01					
69	1.00	3.84	-1.05	4.89	98	1.00	18.67	-1.17	4.95					
70	0.47	6.44	1.49	4.96	99	0.48	21.30	1.24	5.02					
70	1.00	3.78	-1.12	4.89	INT. VIN 11 → 21 V									
71	0.47	6.38	1.42	4.96	99	1.00	18.66	-1.19	4.95					
71	1.00	3.71	-1.19	4.89	100	0.33	23.45	3.17	5.08					
72	0.48	6.36	1.40	4.96	100	1.00	20.01	-0.12	5.02					
72	1.00	3.74	-1.16	4.90	101	0.26	23.77	3.29	5.13					
73	0.48	6.39	1.43	4.97	101	1.00	19.94	-0.36	5.07					
73	1.00	3.77	-1.14	4.90	102	0.25	23.55	2.92	5.16					
74	0.47	6.37	1.40	4.97	102	1.00	19.64	-0.76	5.09					
INT. ROUT 1.0 → 2.5 Ω					103	0.24	23.10	2.40	5.18					
74	1.00	3.73	-14.41	4.50	103	1.00	19.13	-1.29	5.09					
75	0.52	6.83	-10.70	4.38	104	0.25	22.74	2.02	5.18					
75	1.00	4.70	-12.04	4.16										

IMPORTANT NOTICE

Texas Instruments and its subsidiaries (TI) reserve the right to make changes to their products or to discontinue any product or service without notice, and advise customers to obtain the latest version of relevant information to verify, before placing orders, that information being relied on is current and complete. All products are sold subject to the terms and conditions of sale supplied at the time of order acknowledgment, including those pertaining to warranty, patent infringement, and limitation of liability.

TI warrants performance of its products to the specifications applicable at the time of sale in accordance with TI's standard warranty. Testing and other quality control techniques are utilized to the extent TI deems necessary to support this warranty. Specific testing of all parameters of each device is not necessarily performed, except those mandated by government requirements.

Customers are responsible for their applications using TI components.

In order to minimize risks associated with the customer's applications, adequate design and operating safeguards must be provided by the customer to minimize inherent or procedural hazards.

TI assumes no liability for applications assistance or customer product design. TI does not warrant or represent that any license, either express or implied, is granted under any patent right, copyright, mask work right, or other intellectual property right of TI covering or relating to any combination, machine, or process in which such products or services might be or are used. TI's publication of information regarding any third party's products or services does not constitute TI's approval, license, warranty or endorsement thereof.

Reproduction of information in TI data books or data sheets is permissible only if reproduction is without alteration and is accompanied by all associated warranties, conditions, limitations and notices. Representation or reproduction of this information with alteration voids all warranties provided for an associated TI product or service, is an unfair and deceptive business practice, and TI is not responsible nor liable for any such use.

Resale of TI's products or services with statements different from or beyond the parameters stated by TI for that product or service voids all express and any implied warranties for the associated TI product or service, is an unfair and deceptive business practice, and TI is not responsible nor liable for any such use.

Also see: [Standard Terms and Conditions of Sale for Semiconductor Products](http://www.ti.com/sc/docs/stdterms.htm). www.ti.com/sc/docs/stdterms.htm

Mailing Address:

Texas Instruments
Post Office Box 655303
Dallas, Texas 75265